/****************************************************************************
**
** Copyright (C) 2016 The Qt Company Ltd.
** Contact: https://www.qt.io/licensing/
**
** This file is part of the Qt3D Editor of the Qt Toolkit.
**
** $QT_BEGIN_LICENSE:GPL-EXCEPT$
** Commercial License Usage
** Licensees holding valid commercial Qt licenses may use this file in
** accordance with the commercial license agreement provided with the
** Software or, alternatively, in accordance with the terms contained in
** a written agreement between you and The Qt Company. For licensing terms
** and conditions see https://www.qt.io/terms-conditions. For further
** information use the contact form at https://www.qt.io/contact-us.
**
** GNU General Public License Usage
** Alternatively, this file may be used under the terms of the GNU
** General Public License version 3 as published by the Free Software
** Foundation with exceptions as appearing in the file LICENSE.GPL3-EXCEPT
** included in the packaging of this file. Please review the following
** information to ensure the GNU General Public License requirements will
** be met: https://www.gnu.org/licenses/gpl-3.0.html.
**
** $QT_END_LICENSE$
**
****************************************************************************/
import QtQuick 2.5
import com.theqtcompany.SceneEditor3D 1.0

Item {
    property string propertyName: ""
    property string entityName: editorContent.selectedEntityName
    property var component: null
    property int componentType: EditorSceneItemComponentsModel.Unknown
    property var componentValue: (component === null) ? "" : component[propertyName]

    // Inserts property change command to undostack (which also executes the command).
    // oldValue parameter can be used by subclass to control the undo value in cases
    // where temporary property changes are done before editing is finished.
    function handleEditingFinished(newValue, oldValue) {
        if (!oldValue)
            oldValue = component[propertyName]
        if (oldValue !== newValue) {
            editorScene.undoHandler.createChangePropertyCommand(
                        entityName, componentType,
                        propertyName, newValue, oldValue, true)
        }
    }

    // Changes the property without pushing it to undo stack.
    // Useful for changes that can be rejected, such as previewing color selection from ColorDialor.
    function doTemporaryPropertyChange(newValue) {
        var oldValue = component[propertyName];
        if (oldValue !== newValue) {
            // Execute temporary command
            editorScene.undoHandler.createChangePropertyCommand(
                        entityName, componentType,
                        propertyName, newValue, oldValue, false)
        }
    }
}

